
1 [bookmark: _Toc122434485][bookmark: _Toc157694607]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk49 related to the title of this CR.
Contact:	Olivier Genoud
	olivier.genoud@etsi.org
2 [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc157694608]Corrections required
	Function name
	f_MCX_IMS_ServiceAuthorization

	Reason for change
	1. Timer t_WaitForSubscribe_GROUPKEY is never started
2. Wrong condition is used at step 2a3

	Summary of change
	1. Timer t_WaitForSubscribe_GROUPKEY started, when SUBSCRIBE for group document has been received without entry requesting the GMK
2. Condition corrected at step 2a3

	TTCN module
	MCX_SIP_Registration



	  function f_MCX_IMS_ServiceAuthorization(MCX_ServiceType_Type p_ServiceType) runs on MCX_IMS_PTC
  { /* @sic R5-215730: p_Procedure removed sic@ */
    var InternetProtocol_Type v_Protocol := f_IMS_PTC_ImsInfo_GetTransportProtocol();
    var charstring v_MCSUEID := f_SIP_InstanceId();
    var charstring v_GroupId := f_MCX_MCService_GroupID_Group_A(p_ServiceType);
    var MCX_SecurityParameters_Type v_SecurityParameters;
    var MCX_Coordination_MSG v_Dummy;
    var MCX_DataRequestWithXmlBodies_Type v_DataRequestWithXmlBodies_ByRef;
    var MCX_DataRequestWithXmlBodies_Type v_DataRequestWithXmlBodies;
    var template (value) MessageHeader v_MessageHeader_NOTIFY_CONFIG;
    var template (value) MessageHeader v_MessageHeader_NOTIFY_GROUPCONFIG;
    var template (value) MCX_XcapDiff_Type v_XcapDiff;
    var template (value) MIKEY_Message v_MIKEY_Message;
    var ResourceLists_Type v_ResourceLists;
    var ResourceLists_ListSequence_Type v_ResourceLists_SequenceList;
    var boolean v_WaitForRegister := true;
    var boolean v_WaitForPublish := true;
    var boolean v_WaitForSubscribe_CONFIG := true;
    var boolean v_WaitForSubscribe_GROUPCONFIG := true;
    var boolean v_WaitForSubscribe_GROUPKEY := tsc_MCX_Security;                        // group key subscription happens only if security is enabled
    var boolean v_WaitFor200OK_CONFIG := false;
    var boolean v_WaitFor200OK_GROUPCONFIG := false;
    var IMS_PTC_DialogIndex_Type v_DialogIndex_GMS := xcapDiffDialogGMS1;
    var IMS_PTC_DialogIndex_Type v_DialogIndex_GROUPCONFIG;
    var template (omit) IMS_PTC_DialogIndex_Type v_DialogIndex_GROUPKEY := omit;
    var template (present) ResourceLists_ListSequence_Type v_ResourceListsSequence_GMS;
    var SipUrl v_RequestUri_CMS := f_SIP_ParseURI("sip:" & tsc_MCX_CMS_Hostname);       // 36.579-1 Table 5.5.2.14-1 condition CONFIG           @sic R5s220869 Change 6: f_SIP_ParseURI instead of cr_SipUri_HostPort sic@
    var SipUrl v_ContactUri_CMS := v_RequestUri_CMS;                                    // 36.579-1 Table 5.5.2.17.1.2-1 condition CONFIG
    var SipUrl v_RequestUri_GMS := f_SIP_ParseURI("sip:" & tsc_MCX_GMSURI);             // 36.579-1 Table 5.5.2.14-1 condition GROUPCONFIG      @sic R5s220869 Change 6: f_SIP_ParseURI instead of cr_SipUri_HostPort sic@
    var SipUrl v_ContactUri_GMS := v_RequestUri_GMS;                                    // 36.579-1 Table 5.5.2.17.1.2-1 condition GROUPCONFIG  @sic R5w220205 proposal 2.3-7a, R5-223945 Change 5a sic@
 
    timer t_WaitForSubscribe_GROUPKEY := 5.0;

    while (v_WaitForRegister or v_WaitForPublish or v_WaitForSubscribe_CONFIG or v_WaitForSubscribe_GROUPCONFIG or v_WaitForSubscribe_GROUPKEY or v_WaitFor200OK_CONFIG or v_WaitFor200OK_GROUPCONFIG) {
      
      // determine receive template for ResourceLists at the GMS:
      if (v_WaitForSubscribe_GROUPCONFIG and v_WaitForSubscribe_GROUPKEY) {
        v_ResourceListsSequence_GMS := ( f_MCX_ResourceLists_SequenceList_GroupDocGKTP(p_ServiceType), // URI of Group Configuration document and URI of GKTP document in any order
                                         f_MCX_ResourceLists_SequenceList_GroupDocOnly(p_ServiceType), // URI of Group Configuration document
                                         f_MCX_ResourceLists_SequenceList_GKTP_Only(p_ServiceType) );  // URI of GKTP document
      } else if (v_WaitForSubscribe_GROUPCONFIG) {
        v_ResourceListsSequence_GMS := f_MCX_ResourceLists_SequenceList_GroupDocOnly(p_ServiceType);
      } else if (v_WaitForSubscribe_GROUPKEY) {
        v_ResourceListsSequence_GMS := f_MCX_ResourceLists_SequenceList_GKTP_Only(p_ServiceType);
      }
      
      alt {
        // ---------------------------------------------------- Table 5.3.2.3-2  ----------------------------------------------------
        // Table 5.3.2.3-2: step 1a1, 1a2                       REGISTER, 200 OK
        [v_WaitForRegister] a_MCX_SIP_ServiceAuthorization_ReceiveRegisterSend200OK(p_ServiceType) {
          v_WaitForRegister := false;
        }
        // Table 5.3.2.3-2: step 1a3, 1a4 or 1b1, 1b2           PUBLISH, 200 OK
        [v_WaitForPublish] a_MCX_SIP_PocSettings_ReceivePublishSend200OK(p_ServiceType, v_WaitForRegister) {  // IF (v_WaitForRegister) THEN PUBLISH is for service authorization (i.e. shall contain MIKEY message)
          v_WaitForRegister := false;
          v_WaitForPublish := false;
        }

        // ---------------------------------------------------- Table 5.3.2.3-2A ----------------------------------------------------
        // Table 5.3.2.3-2A: step 1, 2                          SUBSCRIBE, 200 OK
        [v_WaitForSubscribe_CONFIG] a_MCX_SIP_ServiceAuthorization_ReceiveSubscribeSend200OK(v_DataRequestWithXmlBodies_ByRef,
                                                                                             p_ServiceType,
                                                                                             -,
                                                                                             xcapDiffDialogCMS,
                                                                                             v_RequestUri_CMS,
                                                                                             v_ContactUri_CMS,
                                                                                             f_MCX_ResourceLists_SequenceList_CONFIG(p_ServiceType, v_MCSUEID)) {
          v_DataRequestWithXmlBodies := v_DataRequestWithXmlBodies_ByRef;
          v_WaitForSubscribe_CONFIG := false;

          f_MCX_SendTrigger(HTTP, CMS);    // send trigger to tell HTTP PTC, that from now on the UE can ask for configuration documents from CMS
        }
        // Table 5.3.2.3-2A: step 3                             NOTIFY
        [] a_MCX_WaitForCoOrdMsg(HTTP, cmr_MCX_Trigger(CMS), v_Dummy) {    // wait for confirmation the HTTP PTC is ready to receive CMS HTTP GET requests (Table 5.3.2.3-2A: step 5, 7, 9)
          v_XcapDiff := f_MCX_XcapDiff_CONFIG(p_ServiceType, v_MCSUEID);
          v_MessageHeader_NOTIFY_CONFIG := f_MCX_SIP_ServiceAuthorization_SendNotify(p_ServiceType, xcapDiffDialogCMS, v_Protocol, v_XcapDiff);
          v_WaitFor200OK_CONFIG := true;
        }
        // Table 5.3.2.3-2A: step 4                             200 OK for NOTIFY
        [v_WaitFor200OK_CONFIG] a_MCX_SIP_ServiceAuthorization_NotifyReceive200OK(v_MessageHeader_NOTIFY_CONFIG, v_Protocol) {
          v_WaitFor200OK_CONFIG := false;
        }

        // ---------------------------------------------------- Table 5.3.2.3-2B or Table 5.3.2.3-2C --------------------------------
        // Table 5.3.2.3-2B/2C: step 1, 2                       SUBSCRIBE, 200 OK
        // NOTE: SUBSCRIBEs of Table 5.3.2.3-2A and Table 5.3.2.3-2C cannot be distinquished in the first place as they have the same message headers and the same request URI
        [v_WaitForSubscribe_GROUPCONFIG or v_WaitForSubscribe_GROUPKEY] a_MCX_SIP_ServiceAuthorization_ReceiveSubscribeSend200OK(v_DataRequestWithXmlBodies_ByRef,
                                                                                                                                 p_ServiceType,
                                                                                                                                 -,
                                                                                                                                 v_DialogIndex_GMS,
                                                                                                                                 v_RequestUri_GMS,
                                                                                                                                 v_ContactUri_GMS,
                                                                                                                                 cr_MCX_ResourceLists(v_ResourceListsSequence_GMS)) {
          v_DataRequestWithXmlBodies := v_DataRequestWithXmlBodies_ByRef;
          v_ResourceLists := v_DataRequestWithXmlBodies.XmlBodies.ResourceLists;
          v_ResourceLists_SequenceList := v_ResourceLists.sequence_list[0].list.sequence_list;

          if (match(v_ResourceLists_SequenceList, {*, f_MCX_ResourceLists_List_Sequence_GroupDoc(p_ServiceType), *} )) {
            v_DialogIndex_GROUPCONFIG := v_DialogIndex_GMS;
            v_WaitForSubscribe_GROUPCONFIG := false;
            f_MCX_SendTrigger(HTTP, GMS);                       // send trigger to tell HTTP PTC, that from now on the UE can ask for group configuration documents from GMS
          }
          if (match(v_ResourceLists_SequenceList, {*, f_MCX_ResourceLists_List_Sequence_GKTP(p_ServiceType), *} )) {
            v_DialogIndex_GROUPKEY := v_DialogIndex_GMS;
            v_WaitForSubscribe_GROUPKEY := false;
            t_WaitForSubscribe_GROUPKEY.stop;
          }
          v_DialogIndex_GMS := fl_MCX_IMS_GetNextDialogIndex(v_DialogIndex_GMS);   /* 'increment' dialog index   @sic R5-212291: use of int2enum sic@ */
          if (v_WaitForSubscribe_GROUPKEY) {
            t_WaitForSubscribe_GROUPKEY.start;
          }
        }
        // Table 5.3.2.3-2B: step 3                             NOTIFY
        [] a_MCX_WaitForCoOrdMsg(HTTP, cmr_MCX_Trigger(GMS), v_Dummy) {    // wait for confirmation the HTTP PTC is ready to receive GMS HTTP GET request (Table 5.3.2.3-2B: step 5)
          v_XcapDiff := f_MCX_XcapDiff_GROUPCONFIG(p_ServiceType, v_GroupId);
          v_MessageHeader_NOTIFY_GROUPCONFIG := f_MCX_SIP_ServiceAuthorization_SendNotify(p_ServiceType, v_DialogIndex_GROUPCONFIG, v_Protocol, v_XcapDiff);
          v_WaitFor200OK_GROUPCONFIG := true;
        }
        // Table 5.3.2.3-2B: step 4                             200 OK for NOTIFY
        [v_WaitFor200OK_GROUPCONFIG] a_MCX_SIP_ServiceAuthorization_NotifyReceive200OK(v_MessageHeader_NOTIFY_GROUPCONFIG, v_Protocol) {
          v_WaitFor200OK_GROUPCONFIG := false;
        }
        // ---------------------------------------------------- Table 5.3.2.3-2C ----------------------------------------------------
        // Table 5.3.2.3-2C: step 2b1                           TIMEOUT
        [] t_WaitForSubscribe_GROUPKEY.timeout {
          v_WaitForSubscribe_GROUPKEY := false;
        }
      }
    }
    f_MCX_WaitForTrigger(HTTP);    // Trigger - HTTP is completed
    
    if (isvalue(v_DialogIndex_GROUPKEY_GROUPCONFIG)) {
      // ---------------------------------------------------- Table 5.3.2.3-2C ----------------------------------------------------
      // Table 5.3.2.3-2C: step 2a3                           NOTIFY

      v_SecurityParameters := f_MCX_IMS_PTC_SecurityParameters_Get();

      v_MIKEY_Message := f_MIKEY_SAKKE_MessageTX_GMK(p_ServiceType, v_SecurityParameters, v_GroupId);
      v_XcapDiff := f_MCX_XcapDiff_GROUPKEY(p_ServiceType, v_GroupId, encvalue(v_MIKEY_Message));
      f_MCX_SIP_XcapDiffNotification(p_ServiceType, valueof(v_DialogIndex_GROUPKEY), v_XcapDiff);
    }
    f_MCX_IMS_PTC_GroupInfo_Add(v_GroupId, v_DialogIndex_GROUPCONFIG, v_DialogIndex_GROUPKEY);
  }




